

# SCIENCE & VIE

MENSUEL

N° 843 DÉCEMBRE 1987

## INCROYABLE ACROBATIE SUR LA "LUNE" DE MARS

**COUP DE FOUDRE  
MONDIAL POUR LA  
SUPRACONDUCTIVITÉ**

**LES JOUETS  
SCIENTIFIQUES**

M 2578 - 843 - 17,00 F



3792578017005 08430

SUISSE 3-FS - CANADA \$3,25 - BELGIQUE 120 FB - ESPAGNE 560 Ptas - MAROC 18 Dh - TUNISIE 2,09 D



# Des graphiques réutilisables

INFORMATIQUE AMUSANTE

**D**e nombreuses fois, dans cette rubrique, nous avons utilisé divers symboles notamment pour la réalisation de jeux. Ce mois-ci, nous avons réuni la majorité d'entre eux dans le programme que nous vous proposons. Il permet de les visualiser directement sous les trois modes que possède l'Amstrad et d'examiner la manière dont ils sont construits.

En effet une fois la figure choisie sélectionnée, le programme fournira directement la position, le nombre et le contenu des symboles nécessaires à sa représentation et ce, directement, sous forme des instructions à réutiliser dans un programme quelconque.

Notons enfin que certains graphiques utilisés au départ sur le Spectrum ont été adaptés ici à

l'Amstrad.

Pour en venir à l'écriture du programme, on remarquera que l'ensemble des données nécessaires à la représentation de chaque illustration a été mémorisé sous forme de DATA. Les premières lignes du programme (lignes 30 à 960) sont donc réservées à cet effet. Nous avons fait précéder chaque groupe de DATA d'une indication, sous forme de REM, de manière à savoir à quel objet il se rapporte afin de simplifier l'étude du programme.

Enfin précisons que la première ligne de DATA de chaque groupe, ne comportant que deux données, permet à l'ordinateur de savoir de combien de symboles graphiques se compose la figure. La première donnée indique le nombre de symboles dans le sens horizontal, la

seconde, dans le sens vertical. Si nous prenons, par exemple, le cas de la locomotive, nous trouvons en ligne 400 : DATA 6,2. Ceci indiquera à l'ordinateur que sa représentation occupe six symboles dans le sens horizontal et deux dans le sens vertical. Les lignes DATA suivantes contiendront les données nécessaires à la redéfinition de chacun de ces symboles.

Ensuite le programme affichera sur l'écran l'ensemble des dessins mémorisés. Cette opération sera assurée par la quadruple boucle FOR... NEXT des lignes 1000 à 1120.

Précisons que les boucles formées à l'aide de P et Q ont pour mission de positionner le dessin sur l'écran et que celles utilisant X et Y contrôlent la position de chaque symbole graphique à l'intérieur d'un même dessin. Notons ici que NBX et NBY seront utilisées pour lire les deux premières données de chaque dessin et prendront donc respectivement pour valeur le nombre de symboles horizontaux

```

10 MODE 1:BORDER 1:PAPER 0:PEN 1:CLS
14 REM *****
15 REM *
20 REM *          DOMINO
25 REM *
26 REM *****
30 DATA 1,2
40 DATA 255,225,225,225,229,225,225,255
50 DATA 225,233,225,225,229,225,255,0
54 REM *****
55 REM *
60 REM *          SOUS-MARIN
65 REM *
66 REM *****
70 DATA 3,2
80 DATA 0,0,0,0,0,0,195
90 DATA 12,8,30,63,63,63,255,255
100 DATA 0,0,0,0,0,0,192,252
110 DATA 175,159,159,159,167,195,0,0
120 DATA 255,255,255,255,255,255,63
130 DATA 254,255,255,255,254,252,192,0
134 REM *****
135 REM *
140 REM *          VEDETTE
145 REM *
146 REM *****
150 DATA 3,1
160 DATA 0,1,1,15,15,255,127,63
170 DATA 0,207,216,240,254,255,255,255
180 DATA 0,128,0,124,96,255,248,224
184 REM *****
185 REM *
190 REM *          HELICOPTERE GAUCHE
195 REM *
196 REM *****
200 DATA 3,2
210 DATA 0,127,0,0,14,49,65,129
220 DATA 8,227,20,8,8,28,188,188
230 DATA 0,255,0,0,0,0,1,6
240 DATA 129,129,129,129,127,127,63,31
250 DATA 254,254,255,255,255,255,252
260 DATA 6,11,3,143,255,252,224,0
264 REM *****
265 REM *
270 REM *          HELICOPTERE DROITE
275 REM *
276 REM *****
280 DATA 3,2
290 DATA 0,255,0,0,0,0,128,96
300 DATA 16,199,40,16,16,56,61,61
310 DATA 0,254,0,0,112,140,130,129
320 DATA 96,176,192,241,255,63,7,0
330 DATA 127,127,255,255,255,255,63
340 DATA 129,129,129,129,254,254,252,248
344 REM *****
345 REM *
350 REM *          POISSON
355 REM *
356 REM *****
360 DATA 2,1
370 DATA 3,31,111,255,127,63,31,3
380 DATA 129,195,247,255,255,247,195,129
384 REM *****
385 REM *
390 REM *          LOCOMOTIVE
395 REM *
396 REM *****
400 DATA 6,2
410 DATA 126,255,255,126,24,24,127,255
420 DATA 0,0,0,14,31,31,255,255
430 DATA 15,12,8,8,8,8,252,255
440 DATA 254,112,32,32,32,32,112,240
450 DATA 0,0,0,1,2,5,10,21
460 DATA 0,16,168,84,170,85,127,127
470 DATA 255,255,255,127,8,34,119,34
480 DATA 255,255,195,153,60,126,60,24
490 DATA 255,255,135,51,121,253,120,48
500 DATA 240,240,240,254,2,68,238,68
510 DATA 255,255,255,255,143,39,112,32
520 DATA 255,255,255,255,241,228,14,4
524 REM *****
525 REM *
530 REM *          WAGON
535 REM *
536 REM *****
540 DATA 4,2
550 DATA 63,119,102,4,4,4,4,70
560 DATA 255,255,115,33,33,33,33,115
570 DATA 255,255,231,66,66,66,66,231
580 DATA 254,247,48,16,16,16,16,49
590 DATA 71,71,71,127,199,147,59,16
600 DATA 255,84,85,84,5,172,255,0
610 DATA 255,68,222,70,246,70,255,0
620 DATA 241,113,241,255,241,228,238,4
624 REM *****
625 REM *
630 REM *          LAMPE A PETROLE
635 REM *
636 REM *****
640 DATA 1,2

```



et verticaux utilisés. Ceci permettra au programme de solliciter le nombre de fois nécessaires la sous-routine 1530, de manière à redéfinir chaque symbole.

L'ensemble des figures ainsi présenté, nous nous attacherons maintenant au contrôle du curseur permettant de sélectionner le graphique choisi. Cette fonction est assurée par les lignes 1130 à 1200 du programme. Les variables XC et YC contrôleront la position horizontale et verticale du curseur et Z permettra de déterminer sous quelle figure il se trouve. Notons que la variable auxiliaire ZP permet d'obtenir un déplacement relativement doux du curseur et autorise également une large tolérance sur sa position par rapport à celle du dessin désiré. Enfin Z sera utilisée pour diriger le programme vers la sous-routine correspondant à la figure choisie (ligne 1210).

Cette sélection effectuée, une procédure de visualisation de la figure choisie sera entamée. Le programme effacera donc l'écran, puis

présentera cette dernière ainsi que, sous forme de lettres, la position des divers symboles graphiques qui la composent. Notons que la position de ces lettres est importante; en effet, elle correspond à l'agencement des instructions "SYMBOL" nécessaires à sa représentation. Par la suite, le programme fournira le contenu de chaque instruction précédé de la lettre relative à sa position; c'est-à-dire sous une forme directement compréhensible par l'ordinateur, sous réserve que la lettre soit remplacée par un numéro. Comme dans un programme une série de redéfinitions de caractères peut porter sur n'importe quels nombres, nous avons choisi ici de figurer ces nombres par des lettres. Pour prendre un exemple, choisissons le cas du domino. Il comporte au total deux symboles graphiques. Si, pourquoi pas?, le premier de ces symboles correspond à 130, A sera donc 130 et B 131.

Cette forme de présentation de l'instruction SYMBOL est donc très



souple et sera acceptée par l'ordinateur quelle que soit la position de départ choisie pour la redéfinition du jeu de caractères et quel que soit leur nombre. Il ne faudra cependant pas oublier d'utiliser l'instruction SYMBOL AFTER xx, avant toute tentative de redéfinition des caractères; xx représente ici le nombre immédiatement inférieur à celui choisi pour la création du premier caractère à redéfinir. Pour les dessins suivants, A, dans notre système de notation, correspondra au numéro du dernier symbole du dessin précédent, plus un.

Reprenons ici un exemple et redéfinissons les symboles correspondant au domino et au sous-

```

650 DATA 24,36,66,153,153,153,153,189
660 DATA 189,153,90,60,255,255,126,24
664 REM *****
665 REM *
670 REM *          OISEAU          *
675 REM *
676 REM *****
680 DATA 2,2
690 DATA 16,72,40,28,14,7,1,128
700 DATA 0,128,96,112,56,28,252,248
710 DATA 99,63,15,7,1,0,0,0
720 DATA 254,253,254,248,192,0,0,0
724 REM *****
725 REM *
730 REM *          CHAMPIGNON      *
735 REM *
736 REM *****
740 DATA 1,2
750 DATA 24,60,126,126,255,255,255,126
760 DATA 0,24,24,60,60,60,60,24
764 REM *****
765 REM *
770 REM *          HACHE VERTICALE  *
775 REM *
776 REM *****
780 DATA 1,2
790 DATA 48,102,240,255,255,240,102,54
800 DATA 6,6,6,6,6,6,6,6
804 REM *****
805 REM *
810 REM *          HACHE HORIZONTALE *
815 REM *
816 REM *****
820 DATA 2,1
830 DATA 24,189,189,24,189,255,126,60
840 DATA 0,255,255,0,0,0,0,0
844 REM *****
845 REM *
850 REM *          FLAME            *
855 REM *
856 REM *****
860 DATA 1,2
870 DATA 9,18,54,54,108,124,254,255
880 DATA 126,255,251,227,231,126,60,24
884 REM *****
885 REM *
890 REM *          FLAME 2          *
895 REM *
896 REM *****
900 DATA 1,2
910 DATA 144,76,100,54,182,127,127,255
920 DATA 126,255,251,227,231,126,60,24
924 REM *****
925 REM *
930 REM *          CHAPEAU          *
935 REM *
936 REM *****
940 DATA 2,1
950 DATA 6,7,7,15,15,15,85,170
960 DATA 96,224,224,240,240,240,85,170
963 REM *****
964 REM *
965 REM * AFFICHAGE DE L'ENSEMBLE DES GRAPHIQUES *
966 REM *
967 REM *****
970 CLS:LET Z=0:LET M=1:SYMBOL AFTER 127:MODE 1
980 LOCATE 1,23:PRINT "POUR DEPLACER LE CURSEUR TAPER ES
FACE."
990 LOCATE 1,25:PRINT "PARAMENTRES DU SYMBOLE CHOISI TA
PER D."
1000 FOR P=3 TO 33 STEP 10
1010 FOR Q=2 TO 17 STEP 5
1020 LET Z=Z+1
1030 ON Z GOSUB 1630,1640,1650,1660,1670,1680,1690,1700,
1710,1720,1730,1740,1750,1760,1770,1780
1040 READ NBX:READ NBY
1050 FOR Y=1 TO NBY
1060 FOR X=1 TO NBX
1070 GOSUB 1530
1080 LOCATE X+P,Y+Q:PRINT CHR$(128)
1090 NEXT X
1100 NEXT Y
1110 NEXT Q
1120 NEXT P
1123 REM *****
1124 REM *
1125 REM * DEPLACEMENT DU CURSEUR SUR L'ECRAN *
1126 REM *
1127 REM *****
1130 LET XC=2:LET YC=5:LET Z=1:LET ZP=1
1140 LOCATE XC,YC:PRINT " "
1150 IF INKEY$=" " THEN LET XC=XC+1:LET ZP=ZP+1:LET Z=1+
INT(ZP/9)
1160 IF UPPER$(INKEY$)="D" THEN GOTO 1210
1170 IF XC=37 THEN LET YC=YC+5:LET XC=2
1180 IF YC=25 THEN GOTO 1130
1190 LOCATE XC,YC:PRINT "+++"
1200 GOTO 1140
1203 REM *****
1204 REM *

```



marin. Supposons que le premier symbole du domino soit logé en 128 ; le second occupera, par voie de conséquence, le symbole 129. Pour le sous-marin on comprendra donc aisément que, lors de l'affichage des instructions SYMBOL, A correspondra à 130, B à 131, C à 132, D à 133, E à 134 et F à 135. Il en sera de même pour toute génération de symboles graphiques succédant à des caractères déjà déterminés ou à un numéro spécifique de redéfinition.

Les lignes 1320 à 1380 permettent d'afficher le dessin sélectionné sous les trois modes que comporte l'ordinateur et ainsi de juger de son aspect. Enfin le programme principal s'achèvera par l'affichage du contenu de chaque instruction SYMBOL. Cette fonction est assurée par les lignes 1390 à 1520. Afin de disposer d'une place suffisante sur l'écran, l'Amstrad passera en mode 2 puis l'affichage sera assuré par la double boucle FOR... NEXT des lignes 1430 à 1490. Reste les deux sous-routines auxquelles le

programme principal fait régulièrement appel.

La première, lignes 1530 à 1620, concerne la redéfinition de chaque caractère graphique des dessins. Notons que nous utiliserons ici toujours le symbole 128 ; nous le redéfinirons donc pour chacune des étapes intervenant dans la formation du dessin.

Dans le cas de la locomotive, par exemple, nous le modifierons 12 fois de suite pour la représenter entièrement. Pour cela, à chaque fois, la sous-routine viendra lire le contenu de la chaîne DATA correspondant au caractère à afficher. Cette opération est effectuée par les lignes 1530 à 1600 de la sous-routine, puis la redéfinition du caractère sera assurée par la ligne 1610.

La seconde sous-routine a pour fonction d'initialiser, donc de sélectionner, le groupe d'instructions DATA que viendra lire la sous-routine précédente en fonction du dessin choisi. Il ne sera donc pas surprenant qu'elle soit exclusive-

ment composée d'instructions RESTORE.

L'utilisation de ce programme est extrêmement simple. Après l'avoir entièrement tapé, puis demandé RUN, l'ensemble des dessins disponibles apparaîtra sur l'écran. Le curseur sera alors placé sous la figure choisie en maintenant la touche ESPACE enfoncée aussi longtemps que nécessaire. Il suffira alors de maintenir la touche D appuyée un court instant pour que l'écran affiche le dessin sélectionné ainsi que la disposition des symboles qui le composent. Il sera alors possible de changer de mode en répondant oui à la question : "Voulez-vous changer de mode (O/N)". On tapera O pour oui et N pour non.

Dès que N sera frappé, la composition de chaque symbole sera affichée sur l'écran. Si la barre d'espace est de nouveau enfoncée, l'ensemble des dessins sera réaffiché. Il sera ainsi possible de sélectionner une seconde figure et ainsi de suite.

**Henri-Pierre Penel**

```

1205 REM * DETERMINATION DU GRAPHIQUE  CHOISI *
1206 REM *
1207 REM *****
1210 ON Z GOSUB 1630,1670,1710,1750,1640,1680,1720,1760,
1650,1690,1730,1770,1660,1700,1740,1780
1213 REM *****
1214 REM *
1215 REM * AFFICHAGE DU GRAPHIQUE CHOISI ET DE LA *
1216 REM * POSITION DES SYMBOLES LE COMPOSANT. *
1217 REM *
1218 REM *****
1220 CLS
1230 IF M=1 THEN LOCATE 10,11:PRINT "DISPOSITONS DES SYM
BOLES:"
1240 READ NBX:READ NBY
1250 FOR Y=1 TO NBY
1260 FOR X=1 TO NBX
1270 GOSUB 1530
1280 LOCATE X+1+(7*(M+1)),Y+5:PRINT CHR$(128)
1290 LOCATE X+1+(7*(M+1)),Y+15:PRINT CHR$(X+NBX*(Y-1)+64
)
1300 NEXT X
1310 NEXT Y
1313 REM *****
1314 REM *
1315 REM * DEMANDE DE CHANGEMENT DE MODE. *
1316 REM *
1317 REM *****
1320 LOCATE 2,23:PRINT "VOULEZ VOUS CHANGER DE MODE ? (
O/N)"
1330 LET K$=UPPER$(INKEY$)
1340 IF K$<>"O" AND K$<>"N" THEN GOTO 1330
1350 IF K$="N" THEN GOTO 1390
1360 LOCATE 2,23:INPUT "QUEL MODE CHOISISSEZ VOUS: 0,1 0
U 2 ";M
1370 IF M<0 OR M>2 THEN GOTO 1360
1380 MODE H:GOTO 1210
1383 REM *****
1384 REM *
1385 REM * AFFICHAGE DU CONTENU DE CHAQUE SYMBOLE *
1386 REM * COMPOSANT LE GRAPHIQUE. *
1387 REM *
1388 REM *****
1390 MODE 2
1400 LET L=64
1410 ON Z GOSUB 1630,1670,1710,1750,1640,1680,1720,1760,
1650,1690,1730,1770,1660,1700,1740,1780
1420 READ NBX:READ NBY
1430 FOR Y=1 TO NBY
1440 FOR X=1 TO NBX
1450 LET L=L+1
1460 GOSUB 1530
1470 LOCATE 4,2*(X+(NBX*(Y-1))):PRINT "SYMBOL ";CHR$( L
);";";A;";";B;";";C;";";D;";";E;";";F;";";G;";";H
1480 NEXT X
1490 NEXT Y
1500 LOCATE 10,25:PRINT "TAPER ESPACE POUR UN AUTRE CHOI
X."
1510 IF INKEY$<>" " THEN GOTO 1510
1520 GOTO 970
1522 REM *****
1523 REM *
1524 REM * SOUS-ROUTINE DE *
1525 REM * GENERATION DU SYMBOLE GRAPHIQUE. *
1526 REM *
1527 REM *****
1530 READ A
1540 READ B
1550 READ C
1560 READ D
1570 READ E
1580 READ F
1590 READ G
1600 READ H
1610 SYMBOL 128,A,B,C,D,E,F,G,H
1620 RETURN
1622 REM *****
1623 REM *
1624 REM * SOUS-ROUTINES D'INITIALISATION *
1625 REM * DES DONNEES DES SYMBOLES GRAPHIQUES. *
1626 REM *
1627 REM *****
1630 RESTORE 20:RETURN
1640 RESTORE 60:RETURN
1650 RESTORE 140:RETURN
1660 RESTORE 190:RETURN
1670 RESTORE 270:RETURN
1680 RESTORE 350:RETURN
1690 RESTORE 390:RETURN
1700 RESTORE 530:RETURN
1710 RESTORE 630:RETURN
1720 RESTORE 670:RETURN
1730 RESTORE 730:RETURN
1740 RESTORE 770:RETURN
1750 RESTORE 810:RETURN
1760 RESTORE 850:RETURN
1770 RESTORE 890:RETURN
1780 RESTORE 930:RETURN

```